
Assumptions
 You need to create non-statistical graphics

– Bar charts

– Scatter plots

– Line plots

 You may or may not have used SAS/GRAPH in the past 
to create these graphs

 You are afraid of statistics

 You are not here to learn detailed syntax.  You are here 
to learn what you can do with the SG procedures

4

Section 1: ODS Graphics for the Non-
Statistician

1.1  What is ODS Graphics?
1.2  Producing Non-Statistical Graphs Using Base SAS 
Procedures
1.3  SAS/GRAPH “SG” Procedures
1.4  Using PROC SGPLOT to Produce Non-Statistical Charts 
and Plots
1.5  Controlling Graph Appearance with SGPLOT
1.6  Using PROC SGPANEL to Display Multiple Cells Per 
Page
1.7  Using PROC SGSCATTER to Produce Multiple Plots Per 
Page
1.8  The ODS Graphics Designer

1.9  Editing Graphs Produced by SG Procedures

1.10  Upcoming Features and Conclusions



3

5

What is ODS Graphics?

6

In the Beginning. . .
All SAS graphics were done with "classic" SAS/GRAPH 
procedures such as PROC GCHART and PROC GPLOT 

proc gchart data=orders;
vbar region;

title 'Back Orders by 
Region';
run;

proc plot data=profit;
plot amount*date=type;

title 'Salaries and Profit by 
Month';
run;

Classic SAS/GRAPH Procedures
 The Good

– Very flexible and customizable

 The Bad

– Not so easy to use

– Statistical graphs required running a STAT 
procedure to create an output data set and using a 
SAS/GRAPH procedure to graph the output data

8

Example: Creating LOESS Plots
proc loess data=Melanoma;

model Incidences=Year;
ods output OutputStatistics=Results;
run;

symbol1 color=black value=dot;
symbol2 color=black interpol=join value=none;

proc gplot data=Results;
title1 'Melanoma Data with Default LOESS Fit';
plot DepVar*Year Pred*Year/ 

hminor=3 vminor=3 overlay;
run;

Step 1:
Create output 
data set from 
PROC LOESS.

Step 2:
Plot data from 
output data set 
with PROC 
GPLOT.

ODS Graphics: Why?
Designed to make it easier for statistical users to develop 
commonly-used statistical graphics.  For example:

proc sgplot data=Melanoma;
title1 'Melanoma Data with Default LOESS Fit';
loess x=year y=incidences;
run;

Only one step 
is required.

ODS Graphics: What is It?
1. Graphics capabilities added to statistical procedures

2. Graphics capabilities added to some Base SAS 
procedures

3. New features added to SAS/GRAPH:

– "SG" procedures

– Graphics Template Language 

– ODS Graphics Designer

This collection of features is ODS Graphics.

Some Terminology
The following terms are interchangable:

 Classic SAS/GRAPH

 Device-based Graphics

 SAS/GRAPH "G" Procedures

 The following terms are interchangable:

 ODS Graphics

 Statistical Graphics

SAS/GRAPH "SG" Procedures are a component of 
ODS/GRAPHICS

12

SAS/GRAPH "Classic" vs. ODS Graphics: 
Major Differences

 Internal architecture and how graphs are produced

 Support for replaying graphs

 Specification of graphics options

 Support for annotation

How Graphs Are Produced by 
"Classic" Procedures

GRAPH1 
Window

(ODS Listing)

Hardcopy
Device or File

Other ODS
Destinations

Device 
Driver

Replay 
Stored
Graphs

(GREPLAY)

GRSEG 
Catalog Entry

SAS/GRAPH
Procedure

How Graphs Are Produced by 
ODS Graphics

SAS/GRAPH, 
SAS/STAT or 

Base
Procedure

Image File

ODS
Destinations

Device 
Driver

 No catalog entry is created
 No output to GRAPH1 window (ODS Listing)

Architectural Differences
SAS/GRAPH "Classic Procedures" and ODS Graphics 
use different underlying subsystems to create their 
graphs.  Because of this:

 You can’t combine graphs produced by the two 
systems

 PROC GREPLAY cannot be used to replay multiple 
or individual ODS graphs

Specifying Graphics Options
 For "classic" SAS/GRAPH, many options can be 

specified in a GOPTIONS statement.  

 For ODS Graphics, there is no GOPTIONS statement.  
A limited number of options can be specified on the 
ODS GRAPHICS statement.  Many others are 
specified in the procedure

16

goptions hsize=3 inches htext=12pt;

ods graphics/height=200px imagefmt=jpeg;

Annotating Graphs
For "classic" SAS/GRAPH, the Annotate facility can be 
used to add data-based annotation to graphics created 
by SAS/GRAPH procedures (such as labeling points on 
a graph).

ODS Graphics currently do not support the Annotate 
facility.  Annotate support is expected to be available 
in SAS 9.3.

17

18

Not Just for Statisticians!
Although the original purpose of ODS Graphics was 
to make producing statistical graphics easier, it’s also 
a convenient way for producing general purpose plots 
and charts.

ODS Graphics: How to Use It
For statistical and Base SAS procedures

 Use ODS GRAPHICS statement to activate graphics

 Add options to procedure code to generate graphs

 Use ODS GRAPHICS OFF statement to deactivate 
graphics

20

ODS Graphics Example: Base/Stat Procedures
ods graphics on;
proc freq data=orion.back_orders;
tables region/plots=freqplot;
run;

ods graphics off;

ODS Graphics: How to Use It
For SAS/GRAPH "SG" Procedures

 Use SG procedure statements

 No ODS GRAPHICS statement required

22

ODS Graphics Example: "SG" Procedure

proc sgplot data=orion.back_orders;
vbar region;

title 'Back Orders by Region';
run;

ODS Graphics: Output
Output from ODS Graphics can to go to either:

 An image file (default)

– ODS LISTING destination used

– Image file accessed from Results window

– Output displayed in host image viewer

 An ODS destination 

– HTML, PDF, RTF

– Graph is output as either a separate image (HTML) 
or an embedded image (RTF, PDF) in the ODS 
result file

– Specify destination on ODS statement

– Output document can be viewed in Results Viewer

24

Output to an Image File
Step 1: Submit Code

Output to an Image File
Step 2: Expand entry in Results Panel—results include 

text tables and graphs

Graphs 
(as image 
files)

Text 
table

26

Output to an Image File
Step 3: Double click on graph entry in results.  Graph 

opens in separate viewing application.

Graphs are 
created and 
stored as 
.png files by 
default



14

27

Output to a HTML File
Step 1: Submit Code

28

Output to a HTML File
Step 2: HTML page is created and automatically opened 
in SAS Results Viewer window

HTML page 
contains 
tables and 
graphs

SAS/GRAPH "Classic" vs. ODS Graphics
Output from "Classic" SAS/GRAPH procedures cannot be 
combined in SAS with output from ODS Graphics.

29

SAS/GRAPH Classic Procedures ODS Graphics

Output goes to GRAPH1 window or 
other destinations (including ODS)

Output to image file or ODS 
document only

Graph created as a GRSEG entry in 
a SAS catalog

No catalog entries created

GREPLAY procedure replays
graphs stored in catalogs

No GREPLAY procedure

Annotate facility available to add 
elements to existing graphs

No Annotate facility (coming in SAS
9.3)

GOPTIONS statement sets general 
graphics options

GOPTIONS statement not used.  

30

Some Destinations Supported by ODS 
Graphics
Destination Viewer Graphics File Types

HTML Web browser PNG (default), GIF, JPEG, …

RTF Word processor, such 
as Microsoft Word

Embedded in RTF file

PS PostScript viewer, 
such as GSview

Embedded in PostScript file

PDF PDF viewer, such as 
Adobe Reader

Embedded in PDF file

LATEX PostScript or PDF 
viewer after compiling 
LaTeX file

PostScript (default), EPSI, 
GIF, JPEG, PDF, PNG

LISTING Default viewer in your 
system for file type

PNG (default), GIF, BMP, DIB, 
EMF, EPSI, JFIF, JPEG, …

30

Some Recommended Styles
Style Description

DEFAULT Color style intended for general-purpose work. 
This is the default for the HTML destination.

STATISTICAL Color style recommended for output in Web pages 
or color print media. This is the style used in 
SAS/STAT 9.2 documentation.

ANALYSIS Color style with a somewhat different appearance 
from STATISTICAL.

JOURNAL and 
JOURNAL2

Gray-scale and pure black-and-white styles, 
respectively. Recommended for graphs in 
black-and-white publications.

RTF Used to produce graphs to insert into a Microsoft 
Word document or a Microsoft PowerPoint slide.

Producing Non-Statistical Graphs Using Base 
SAS Procedures

34

Base SAS Procedures that Produce Graphics
ODS Graphics can be used with the following Base SAS 
procedures:

 CORR

 FREQ

 UNIVARIATE

These procedures produce both statistical and non-
statistical graphics. This presentation will focus on the 
non-statistical charts and plots that PROC FREQ and 
PROC UNIVARIATE can produce.

Producing Graphics with Base Procedures
For statistical and Base SAS procedures

1. Use ODS GRAPHICS ON statement to activate 
graphics

2. Add options or statements to procedure code to 
generate specific graphs

3. Use ODS GRAPHICS OFF statement to deactivate 
graphics

36

PROC FREQ Graphics with One-Way Tables
If you activate graphics, but do not specify a specific plot 
with a one-way frequency table request, both frequency 
and cumulative frequency plots are generated:

ods graphics on;
proc freq data=orion.back_orders;
tables region;

run;
ods graphics off;

PROC FREQ Graphics with Two-Way Tables
If you activate graphics, but do not specify a specific plot 
with a two-way frequency table request, a single graph is 
generated containing a frequency plot for each variable.  
No cumulative frequency plot is produced.

ods graphics on;
proc freq 

data=orion.employee_payroll;
tables employee_gender*

marital_status;
run;
ods graphics off;

38

Using the PLOTS= Option with PROC FREQ
You can use the PLOTS= option to request specific types 
of plots, including various statistical plots.  By using the 
PLOTS= option, you can specify some appearance 
options.

ods graphics on;
proc freq data=orion.employee_payroll;

tables marital_status/plots=freqplot
(orient=horizontal 
scale=percent);

run;
ods graphics off;

Creating a Dot Plot with PROC FREQ
The example below requests a dot plot for the frequency 
plot only.  

ods graphics on;
proc freq data=orion.orders;

tables region/plots(only)=freqplot 
(type=dotplot orient=vertical);

run;
ods graphics off;

40

Graphics with PROC UNIVARIATE
To produce graphics with PROC UNIVARIATE, you 
must explicitly supply a statement requesting a graph.  
The HISTOGRAM statement produces histograms.

ods graphics on;
proc univariate 
data=orion.employee_payroll;
var salary;
histogram salary;
run;

ods graphics off;

Graphics with PROC UNIVARIATE
Various options can be used on the HISTOGRAM 
statement to control the appearance of the plot.

ods graphics on;
proc univariate 
data=orion.employee_payroll;
var salary;
histogram salary/barlabel=count

normal(color=blue)
grid;

run;
ods graphics off;

Overview of SAS/GRAPH "SG" Procedures

44

ODS GRAPHICS with SAS/GRAPH
Several new procedures and features that implement 
ODS Graphics have been added to SAS/GRAPH.  
They are:

 PROC SGPLOT

 PROC SGPANEL

 PROC SGSCATTER

 PROC SGRENDER

 Graphics Template Language (GTL)

The "SG" Procedures
 SGPLOT, SGPANEL, and SGSCATTER Procedures

– designed to create commonly used graphs quickly. 

– produce many of the same types of graphs as the 
original SAS/GRAPH procedures such as GPLOT 
and GCHART, but use the ODS architecture

 SGRENDER and the Graphics Template Language 

– used to produce more complex, customized graphs

– steeper learning curve

46

SGPLOT vs. Graph Template Language

proc template;
define statgraph sgdesign;
dynamic _ORDER_MONTH;
begingraph;
layout lattice _id='lattice' / 

columndatarange=data rowdatarange=data;
layout overlay _id='overlay' /;
barchart _id='bar' x=_ORDER_MONTH / name='bar' stat=freq;
endlayout;

endlayout;
endgraph;
end;
run;
proc sgrender data=ORION.ORDER_SUMMARY template=sgdesign;
dynamic _ORDER_MONTH="'ORDER_MONTH'n";
run;

proc sgplot data=orion.back_orders;
vbar region;

title "Frequency Count by Region";
run;

SGPLOT

GTL and PROC SGRENDER

Using the New SG Procedures
 ODS GRAPHICS statement is not required, but can be 

used to specify some options

 No GOPTIONS statement

 Symbols and patterns are specified in the procedure—
no SYMBOL or PATTERN statements

 Statements available to control axis and legend 
appearance, but differ from those in "classic" 
SAS/GRAPH

 Titles and footnotes work as in "classic" SAS/GRAPH

Using PROC SGPLOT to Produce Non-
Statistical Charts and Plots

50

The SGPLOT Procedure
PROC SGPLOT can be used to produce the following 
basic plots and charts:

 Simple bar charts

 Stacked bar charts

 Histograms

 Scatter plots

 Series plots

 Overlaid graphs including bar/line charts

 Although the procedure is named SGPLOT, it 
produces both plots and charts.

PROC SGPLOT: Statistical Graphs
In addition, PROC SGPLOT can be used to produce the 
following types of plots and charts:

 Density plots

 Dot plots

 Ellipse plots

 Horizontal and vertical box plots

 Loess plots

 Spline curves

 Regression plots

51

 For this presentation, we will focus on the non-
statistical plots listed on the previous slide.

52

Terminology for ODS Graphics
Term Meaning

plot any type of plot or chart, such as a scatter plot, bar chart, etc.

cell area containing one plot, or multiple overlaid plots

graph a collection of one or more cells

graph

cell

plot

multiple plots 
overlaid

General SGPLOT Syntax
 PROC SGPLOT statement – invokes the procedure 

and specifies input data set

 Plot statement – specifies type of graph, variables, 
and options.

 Axis statements – control appearance of axes 
(optional)

 KEYLEGEND statement – controls appearance of 
legend (optional)

 Multiple plot statements can be used to overlay 
multiple plots on the same axes.

54

Vertical Bar Chart
proc sgplot data=orion.back_orders;
vbar region;

title "Frequency Count by Region";
run;

Horizontal Bar Charts
proc sgplot data=orion.back_orders;
hbar region / response=number_of_orders
stat=sum;

title "Backorders by Region";
run;

56

Subgrouped Bar Chart
proc sgplot data=orion.employees;

where company=:'Orion';
vbar company / group=department;

title 'Employees By Country/Department';
run;

Histogram
proc sgplot data=orion.employees;

where department='Sales';
histogram salary;

title 'Distribution of Employee Salaries';
run;

58

Overlaying Graphs
proc sgplot data=orion.employees;

histogram salary;
density salary / type=normal;
where department='Sales';

title 'Distribution of Employee Salaries';
run;

Scatter Plot
proc sgplot data=orion.employees;
where company='Orion Italy' and 
department='Sales';
scatter y=salary x=employee_birthdate;
title 'Salary by Birth Date';

run;

60

Grouped Series Plot
proc sgplot data=orion.profit;
where company in ('Orion France','Orion

Italy');
series y=sales x=yymm / group=company;

title 'Sales by Country';
run;

Overlayed Series Plot (Multiple Y Variables)
proc sgplot data=orion.budget;
series y=yr2003 x=month;
series y=yr2004 x=month;

title 'Budget by Month';
run;

62

Overlaid Bar Chart and Line Plot
proc sgplot data=orion.profit;
where company='Orion France' and 

year(yymm)=2004;
vbar yymm / response=sales;
vline yymm / response=cost; 

title 'Sales vs. Cost';
run;

Controlling Graph Appearance with SGPLOT

Using ODS Styles
ods listing style=banker;
proc sgplot data=orion.profit;
where company='Orion France' and 

year(yymm)=2004;
vbar yymm / response=sales;
vline yymm / response=cost; 

title 'Sales vs. Cost';
run;
ods listing style=listing;

Controlling Fills and Labeling Bars
proc sgplot data=orion.back_orders;
vbar region/response=number_of_orders

fillattrs=(color="verylightred")
datalabel;

title "Back Orders by Region";
run;

Overlaid Bars with Transparency
proc sgplot data=orion.budget;

vbar month/response=yr2003
fillattrs=(color=red)
transparency=.7;

vbar month/response=yr2004 
fillattrs=(color=blue)
transparency=.7 barwidth=.5;

run;

68

Controlling Lines and Markers
proc sgplot data=orion.budget;
series y=yr2003 x=month/ markers 

lineattrs=(color=red thickness=3) 
markerattrs=(color=black 
symbol=circlefilled
size=8);

title 'Budget by Month';
run;

Controlling Axis Attributes
proc sgplot data=orion.profit;
where (company='Orion Australia' or 

company='Orion Italy') and 
year(yymm)=2003;

series y=sales x=yymm/ markers group=company;
xaxis fitpolicy=rotate 

tickvalueformat=monname. 
grid label='Month';
yaxis label='Total Sales' grid;

title '2003 Sales';
run;

70

Additional Axis Options
proc sgplot data=orion.profit;
where company in ('Orion Australia' 'Orion 

Italy') and year(yymm)=2003;
series y=sales x=yymm/ markers group=company;
xaxis display=(nolabel) 

tickvalueformat=monname3.
offsetmin=0 offsetmax=0 grid;

yaxis values=(40000 to 240000 by 40000) grid
label='Total Sales'; 

title '2003 Sales';
run;



36

71

Adding Top and Right Axes
proc sgplot data=orion.budget;
series y=yr2003 x=month;
series y=yr2004 x=month/x2axis y2axis;
xaxis type=discrete;
yaxis display=(nolabel) grid offsetmax=.1;
x2axis type=discrete display=(nolabel);
y2axis display=(nolabel) values=(1e6 to 3e6 by

500000)offsetmax=.1;
title 'Budget by Month';
run;

Controlling the Legend
proc sgplot data=orion.budget;
series y=yr2003 x=month;
series y=yr2004 x=month;
xaxis type=discrete;
keylegend / location=inside down=2  

position=topleft;
title 'Budget by Month';
run;

Adding a Text Box With INSET Statement
proc sgplot data=orion.budget;
series y=yr2003 x=month / name='2003';
series y=yr2004 x=month / name='2004';
xaxis type=discrete;
inset  'This plot illustrates the monthly' 

'budget values for 2003 and 2004'/
position=bottom border  
title='Explanation: ';

title 'Budget by Month';
run;

74

The ODS GRAPHICS Statement
You can specify options on the ODS GRAPHICS statement 
to control some aspects of your graphics output. Note that 
you do not need to use the ODS GRAPHICS statement 
with the SG procedures unless you want to specify options. 
Some of the options that you can specify are:

 Height and width of the graph

 Format (file type) and name of the graphics file

 Whether a border is drawn around the graph

Using the ODS GRAPHICS Statement
ods graphics / height=200px width=500px 
imagefmt=jpeg;
proc sgplot data=orion.back_orders;
vbar region/response=number_of_orders

fillattrs=(color="verylightred")
datalabel;

title "Back Orders by Region";
run;

76

The ODS GRAPHICS Statement
Options specified on the ODS GRAPHICS statement remain 
in effect until they are reset or the SAS session ends. To 
reset the options, submit the following statement:

ods graphics/reset;

Creating Tool Tips with the IMAGEMAP Option
The IMAGEMAP option is used with the HTML destination 
and controls the generation of data tips. Data tips are 
pieces of explanatory text that appear when you mouse-
over the data portions of a graph contained in an HTML 
page. The TIPMAX option specifies the maximum number 
of distinct tooltips permitted before tooltips are disabled.

77

ods graphics / imagemap tipmax=1500;

ods html path='path-for-output' (url=none)
gpath='path-for-image'
file='myfile.html' style=analysis;

. . . code that generates ODS GRAPHICS image . . .

ods html close;

78

Section 1: ODS Graphics for the Non-
Statistician

1.1  What is ODS Graphics?
1.2  Producing Non-Statistical Graphs Using Base SAS 
Procedures
1.3  SAS/GRAPH “SG” Procedures
1.4  Using PROC SGPLOT to Produce Non-Statistical Charts 
and Plots
1.5  Controlling Graph Appearance with SGPLOT

1.6  Using PROC SGPANEL to Display Multiple 
Cells Per Page
1.7  Using PROC SGSCATTER to Produce Multiple Plots Per 
Page
1.8  The ODS Graphics Designer
1.9  Editing Graphs Produced by SG Procedures

1.10  Upcoming Features and Conclusions



40

79

Using PROC SGPANEL to Display Multiple 
Cells Per Page

80

PROC SGPANEL
PROC SGPANEL produces graphs similar to PROC 
SGPLOT, but creates separate cells for each value of 
a categorical variable or crossing of multiple categorical 
variables. You can control the placement of the cells on 
a graph.

General SGPANEL Syntax
 PROC SGPANEL statement – invokes the procedure 

and specifies input data set

 Plot statement – specifies type of graph, variables, 
and options.

 PANELBY statement – specifies the classification 
variables and the arrangement of cells on the page.

 Axis statements – control appearance of axes 
(optional)

 KEYLEGEND statement – controls appearance of 
legend (optional)

82

PROC SGPANEL: Plot Statements
 The plot statements supported by PROC SGPANEL 

are the same as those supported by PROC SGPLOT. 
They include HBAR, VBAR, HISTOGRAM, SCATTER, 
SERIES, and others.

 As with PROC SGPLOT, multiple plot statements 
produce overlay graphs.

 The same options used to control plot appearance 
with PROC SGPLOT (such as LINEATTRS, 
FILLATTRS, etc.) can be used with PROC SGPANEL.

The PANELBY Statement
The PANELBY statement lists one or more categorical 
variables.

proc sgpanel data=orion.profit;
where company in ('Orion Belgium' 

'Orion Germany' 
'Orion France' 
'Orion Italy');

panelby company;
series y=sales x=yymm;
title 'Monthly Sales';
run;

84

PANELBY Statement Options
Options on the PANELBY statement specify the 
arrangement of cells

proc sgpanel data=orion.profit;
where company in ('Orion Belgium‘

'Orion Germany''Orion France' 
'Orion Italy');

panelby company/layout=columnlattice onepanel;
series y=sales x=yymm;
title 'Monthly Sales';
run;

Additional PANELBY Statement Options
proc sgpanel data=orion.profit;
where company in ('Orion Belgium' 

'Orion Germany''Orion France' 
'Orion Italy');

panelby company/layout=columnlattice 
colheaderpos=bottom
onepanel 
noborder       
novarname;

series y=sales x=yymm;
title 'Monthly Sales';
run;

86

Multiple PANELBY Variables
proc sgpanel data=orion.customer_orders;
panelby order_type customer_gender;
vbar customer_country;
title 'Orders by Country, Order Type, 

Gender';
run;



44

87

Removing Borders Between Cells
proc sgpanel data=orion.customer_orders;
panelby customer_gender / noborder    

spacing=30 novarname;
vbar customer_country;
title 'Orders by Country, Gender';
run;

88

Section 1: ODS Graphics for the Non-
Statistician

1.1  What is ODS Graphics?
1.2  Producing Non-Statistical Graphs Using Base SAS 
Procedures
1.3  SAS/GRAPH “SG” Procedures
1.4  Using PROC SGPLOT to Produce Non-Statistical Charts 
and Plots
1.5  Controlling Graph Appearance with SGPLOT
1.6  Using PROC SGPANEL to Display Multiple Cells Per 
Page
1.7  Using PROC SGSCATTER to Produce Multiple 
Plots Per Page
1.8  The ODS Graphics Designer
1.9  Editing Graphs Produced by SG Procedures

1.10  Upcoming Features and Conclusions



45

89

Using PROC SGSCATTER to Produce Multiple 
Plots Per Page

90

PROC SGSCATTER
PROC SGSCATTER produces multiple scatter or 
line plots on a page, with the cells reflecting different 
combinations of plot variables.

General SGSCATTER Syntax
 PROC SGSCATTER Statement – invokes the 

procedure and specifies input data set

 PLOT, COMPARE, MATRIX statement – specifies 
type of graph, variables, and options.

 Axis statements are not supported

 Legend is controlled by options in PLOT statement

Note that:

– Only certain types of plots can be produced

– Plot type is specified as an option on the PLOT, 
COMPARE, or MATRIX statement.

PLOT Statement
proc sgscatter data=orion.budget;
where month le 6;
plot (yr2003 yr2004 yr2005 yr2006)*month;
title 'Monthly Budget by Year';
title2 '(First 6 Months)';
run; 

Arranging Multiple Cells on a Page
proc sgscatter data=orion.budget;
where month le 6;
plot (yr2003 yr2004 yr2005)*month/join 

columns=1 uniscale=y;
title 'Monthly Budget by Year';
title2 '(First 6 Months)';
run; 

COMPARE Statement  – Shared Axes
proc sgscatter data=orion.budget;
where month le 6;
compare x=month y=(yr2003 yr2004 yr2005) / 

join spacing=10;
title 'Monthly Budget by Year';
title2 '(First 6 Months)';
run; 

MATRIX Statement  – Scatterplot Matrices
proc sgscatter data=orion.heart;
where ageatstart>60;
matrix weight diastolic systolic/group=sex

diagonal=(histogram normal);
title 'Comparison of Weight, Systolic, and 
Diastolic';
run;

SGPANEL vs. SGSCATTER
What’s the difference?

 SGPANEL creates a separate cell for each value of a 
categorical variable or variables.

– Example: Sales by Month with a separate cell for 
each country

 SGSCATTER creates cells for different combinations 
of variables

– Example: Sales by Month and Cost by Month (for 
all countries), on the same page

The ODS Graphics Designer

ODS Graphics Designer
The ODS Graphics Designer provides an interactive 
interface for creating the same types of graphs produced 
by the SGPLOT, SGPANEL, and SGRENDER 
procedures. It also allows creation of some types of 
graphs that cannot be created with the above procedures.

To invoke the ODS Graphics Designer, submit the 
following statement from your SAS Editor:

The ODS Graphics Designer comes up in a separate 
application window.

%sgdesign;

Graphics Designer Main Window

Creating a Graph Using the Designer
 Select a graph type from the Graph Gallery. Graph is 

displayed using default data sets and variables.

 Fill in popup form to specify your data set, variables, 
and options. Updated graph is displayed.

 Add additional plots and elements by dragging them 
from the Elements Pane onto the graph

or

 Add additional panels to the graph and then drag plots 
and elements into the panels.

Select a Graph from Graph Gallery
 Graph is created with default data

 Data menu is displayed

Choose Data Set and Variables

104

Add Title and Right Click for Options

After Changes

106

Drag Legend On To Graph

Add Another Panel to Graph
Select Row from Insert pull-down.

108

New Row Is Added

Drag a Plot Into Empty Row & Specify Details

110

Overlay Another Plot in Panel
Drag Plot Layer into Panel and Fill in Details.

Plot After Adding Overlay

112

Right Click Plot and Add Enhancements

Save Graph to Gallery

114

Specify Location and Name, then Save

We will come back to 
the graph later…..

Creating Paneled Graphs
Select Panel tab from Gallery and choose graph type.

116

Fill in Panel and Plot Variables

Paneled Graph

118

Some Other Graphs in the Gallery

Click "My Graphs" Tab to See Saved Graphs

120

Can I Save Code Generated by the Designer?
 Good news – code can be viewed and saved

 Bad news – code is Graphics Template Language and 
PROC SGRENDER, but…

 Good news – the code is a good way to learn GTL!

Click View Code to See Generated Code

122

Using Saved Layouts as Templates for Other 
Graphs
 Double-click graphs from Gallery window, and change 

parameters as needed

 Modify and run the generated GTL code with PROC 
SGRENDER

 Use PROC SGDESIGN

– You can change input data set and plot variables in 
the PROC SGDESIGN code.  

Using PROC SGDESIGN
This program replays the saved graph, specifying a 
different grouping variable for the bar chart.

proc sgdesign sgd='globalforum2011.sgd';
dynamic _order_type='supplier';
run;

Editing Graphs Produced by SG Procedures

126

Purpose of the ODS Graphics Editor
 Primary purpose – make simple, ad-hoc changes to 

graphs created by SG and other procedures.

 Add simple annotation to graphs (text, lines, symbols)

 Data components (including tick labels) cannot be 
modified.

Enabling the Editing of ODS Graphics
You can use the ODS Graphics Editor to make edits to 
graphs created with ODS graphics.  The ODS Editor 
cannot be used to edit graphcs created with "classic" 
SAS/GRAPH procedures.

To enable use of the editor, submit the following 
statement: 

ods listing sge=on;

Creating an Editable Graph
When a SG procedure is run and SGE=ON is specified, 
the results window shows two graphs – one is an image 
file, and the second is the editable graph.

image file

editable 
graph

Opening the Editable Graph
Double-clicking on the editable graph opens it in the 
ODS Graphics Editor.

Editing the Graph
Right clicking on a graph component brings us a dialog 
you can use to modify the component.

Graph After Editing

132

Saving the Edited Graph
Click on File  Save as….

Saving the Edited Graph
You can save the edited graph as either:

 A .png (Portable Network Graphics) file, which can be 
printed or inserted into other documents

 A .sge file, which can be opened and re-edited with 
the ODS Graphics Editor

 The ODS Graphics Editor does not generate SAS 
code.

Upcoming Features and Conclusions

136

Upcoming ODS Graphics Features in SAS 9.3
 Clustered groups for bar charts

 Fill pattern "skins" for bar charts

 Bubble plots

 High-Low Plots

 Pie Charts

 Attribute maps

 Annotation

 Maps (experimental)

 Default output to web page instead of image file

Clustered Groups and "Skins" for Bar Charts

138

Bubble Plots

High-Low Plots

140

Pie Charts

Annotate

142

Some Considerations
You cannot use SAS to combine ODS Graphics with 
graphs produced by "classic" SAS/GRAPH procedures

ODS Graphics cannot produce some types of graphs that 
"classic" SAS/GRAPH procedures can, such as:

 Pie and donut charts

 Radar charts

 3D bar charts

 Maps

 Tile charts

ODS Graphics does not (yet) support Annotate

continued...

Some Considerations
ODS Graphics can produce the following types of graphs 
that "classic" procedures cannot, including:

 Density plots

 Ellipse plots

 Vector plots

 Horizontal box plots

ODS Graphics can produce the following types of graphs 
much more easily than "classic" SAS/GRAPH procedures:

 Overlay plots

 Panel graphs 

 Plot matrices

144

Which Should I Choose?
 If you are already creating graphs for a specific 

application using "classic" SAS/GRAPH procedures, 
continue to use them (they won’t go away).

 If your graphs require complex annotation, use 
"classic" SAS/GRAPH procedures.

 Consider using SG procedures for new applications 
that do not require complex annotation.

